

2

NPS69-88-003

NAVAL POSTGRADUATE SCHOOL

Monterey, California

DTIC FILE COPY

AD-A202 864



A Triangular Thin Shell Element for the
Linear Analysis of Stiffened Composite
Structures

Edward Wilson and Gilles Cantin

April 1988

Final Report for period Jan 1987 - Jun 1987

DTIC
ELECTE

24 JAN 1989

E

Approved for public release; distribution unlimited
Prepared for:

David Taylor Naval Ship R&D Center
Bethesda, MD 20084-5000

89 1 23 19 8

Unclassified

SECURITY CLASSIFICATION OF THIS PAGE

REPORT DOCUMENTATION PAGE

1a. REPORT SECURITY CLASSIFICATION Unclassified			1b. RESTRICTIVE MARKINGS		
2a. SECURITY CLASSIFICATION AUTHORITY			3. DISTRIBUTION AVAILABILITY OF REPORT Approved for public release; distribution unlimited		
2b. DECLASSIFICATION/DOWNGRADING SCHEDULE					
4. PERFORMING ORGANIZATION REPORT NUMBER(S) NPS69-88-003			5. MONITORING ORGANIZATION REPORT NUMBER(S)		
6a. NAME OF PERFORMING ORGANIZATION Naval Postgraduate School		6b. OFFICE SYMBOL (if applicable) Code 69		7a. NAME OF MONITORING ORGANIZATION Naval Postgraduate School	
6c. ADDRESS (City, State, and ZIP Code) Monterey, California 93943-5000			7b. ADDRESS (City, State, and ZIP Code) Monterey, California 93943-5000		
8a. NAME OF FUNDING/SPONSORING ORGANIZATION David W. Taylor R&D Center		8b. OFFICE SYMBOL (if applicable)		9. PROCUREMENT INSTRUMENT IDENTIFICATION NUMBER N000167087 WR7-0062	
8c. ADDRESS (City, State, and ZIP Code) Bethesda, MD 20084-5000			10. SOURCE OF FUNDING NUMBERS		
			PROGRAM ELEMENT NO. 62323N;	PROJECT NO. 14120;	TASK NO. RB23S11000
11. TITLE (Include Security Classification) A Triangular Thin Shell Element for the Linear Analysis of Stiffened Composite Structures					
12. PERSONAL AUTHOR(S) Edward Olson and Gilles Cantin					
13a. TYPE OF REPORT Final		13b. TIME COVERED FROM Dec 87 TO Jun 88		14. DATE OF REPORT (Year, Month, Day) April 1988	
15. PAGE COUNT 25					
16. SUPPLEMENTARY NOTATION					
17. COSATI CODES			18. SUBJECT TERMS (Continue on reverse if necessary and identify by block number)		
FIELD	GROUP	SUB-GROUP			
19. ABSTRACT (Continue on reverse if necessary and identify by block number) A three node flat shell element with six engineering displacement degree-of-freedom at each node is developed. The basic formulation allows for the arbitrary location of the reference surface in which the membrane forces and bending moments are fully coupled. The well-known, highly accurate, DKT bending element is combined with a higher order membrane element in order to obtain a consistent formulation. The higher order membrane behavior is obtained by the introduction of three additional normal rotational degree-of-freedom. This report presents a summary of the theoretical steps involved in the development of the element. The accuracy of the element is illustrated by the solution of several standard problems and a comparison of results with other thin shell elements. The FORTRAN 77 listing of the subroutines which form the basic element matrices contain less than 300 statements and is presented in order to illustrate that the computer implementation of the element is relatively simple.					
20. DISTRIBUTION/AVAILABILITY OF ABSTRACT <input checked="" type="checkbox"/> UNCLASSIFIED/UNLIMITED <input type="checkbox"/> SAME AS RPT <input type="checkbox"/> DTIC USERS			21. ABSTRACT SECURITY CLASSIFICATION Unclassified		
22a. NAME OF RESPONSIBLE INDIVIDUAL Edward Wilson			22b. TELEPHONE (Include Area Code) 415-625-0438		22c. OFFICE SYMBOL

DD FORM 1473, 84 MAR

83 APR edition may be used until exhausted.
All other editions are obsolete

SECURITY CLASSIFICATION OF THIS PAGE

U.S. Government Printing Office: 1986-606-243

**A TRIANGULAR THIN SHELL ELEMENT
FOR THE LINEAR ANALYSIS OF
STIFFENED COMPOSITE STRUCTURES**

By

Edward Wilson and Gilles Cantin

ABSTRACT

A three node flat shell element with six engineering displacement degrees-of-freedom at each node is developed. The basic formulation allows for the arbitrary location of the reference surface in which the membrane forces and bending moments are fully coupled.

The well-known, highly accurate, DKT bending element is combined with a higher order membrane element in order to obtain a consistent formulation. The higher order membrane behavior is obtained by the introduction of three additional normal rotational degrees-of-freedom.

This report presents a summary of the theoretical steps involved in the development of the element. The accuracy of the element is illustrated by the solution of several standard problems and a comparison of results with other thin shell elements. The FORTRAN 77 listing of the subroutines which form the basic element matrices contain less than 300 statements and is presented in order to illustrate that the computer implementation of the element is relatively simple.



Accession

NTIS Grant

DTIC TAB

Unannounced

Justification

By

Distribution/

Availability Codes

Dist

Avail and/or
Special

A-1

TABLE OF CONTENTS

ABSTRACT	i
TABLE OF CONTENTS	ii
INTRODUCTION	1
BASIC EQUATIONS	3
BENDING APPROXIMATION - THE DKT ELEMENT	6
MEMBRANE APPROXIMATION	10
COMPUTER PROGRAM IMPLEMENTATION	12
NUMERICAL EXAMPLES	12
Cantileaver Beam Example	13
Twisted Beam	14
Scordelis-Lo Roof	15
Spherical Shell	16
FINAL REMARKS	17
REFERENCES	17
FORTRAN LISTING OF 18 DOF TRIANGULAR SHELL ELEMENT	18

INTRODUCTION

General Background

The use of composite materials allows for the efficient design of many different types of structural systems. One of the major advantages of the material is that different stiffness and strength properties can be obtained in different directions. Therefore, more efficient structures can be obtained since the material can be concentrated in the directions of maximum stresses.

Most of the existing finite element programs do not have sufficient generality to consider such material properties. Also, in the case of thin shell structures very few programs have the ability to consider shells in which the bending and membrane forces are coupled. In addition, problems associated with the modelling of complex shell structures with thin shell elements exist since the classical thin shell formulation does not have stiffness terms associated with the normal rotational degrees of freedom. Therefore, the user of the program is often required to add artificial members to a finite element model in order to avoid numerical instability in the solution of the finite element system. The purpose of this report is to present a new thin shell element which is sufficiently robust to solve the above mentioned problems.

Recent Research

Within the past two years several papers have presented methods which introduce a normal rotation in order to improve the membrane behavior of plane elements. Carpenter, Stolarski and Belytschko present a flat triangular shell element with improved membrane interpolation. [1] They introduced normal mid-side displacements of the constant strain triangle. The normal displacements are eliminated and node rotations are introduced by the use of a cubic constraint function along each side of the triangle. A one point integration method is used in order to eliminate membrane locking within the elements. The element yields very accurate displacements; however, the element is rank deficient and is unstable for certain geometries.

Taylor and Simo applied the same basic approach as presented in reference [1] to improve the membrane behavior of quadrilateral elements [2]. For many problems excellent displacements and stresses are obtained. However, for shell structures such as a twisted beam the displacements become very large as the mesh was refined. In addition, the flat quadrilateral element does not accurately model many common types of shell geometries. Also, the DKQ formulation was used to form the bending stiffness which has proven to be not as accurate as the DKT formulation.

Bergan and Felippa have developed a triangular membrane element with normal rotational degrees-of-freedom [3]. The formulation is based on the "free formulation" and uses the continuum-mechanics definition of rotation. The element passes the patch test and is stable for all applications. The element produces good values of displacements; however, the values for stresses are poor compared to the values obtained from the Taylor quadrilateral [4].

The three elements previously mentioned have not been used for thin shells in which the membrane and bending forces are coupled. Therefore, one of the purposes of this report is to develop an element which has a consistent formulation for both the bending and membrane behavior. In addition, the problems with the instability associated with normal rotational degree-of-freedom will be studied and a simple technique is suggested in order to avoid this problem.

BASIC EQUATIONS - ORTHOTROPIC MATERIALS

The 18 x 18 triangular shell element stiffness matrix for a stiffened composite material as shown in figure 1 can be directly calculated from the following well-known equation:

$$\underline{K} = \int \underline{B}^T \underline{D} \underline{B} dA \quad (1)$$

The 6x6 \underline{D} matrix relates the forces to the deformations which are associated with a differential element of area dA . Including thermal deformations the force-deformation relationship can be expressed by the following matrix equation:

$$\underline{f} = \underline{D} \underline{e} + \underline{f}_0 \quad (2)$$

where

$$\underline{f}^T = [M_{11} \ M_{22} \ M_{12} \ f_{11} \ f_{22} \ f_{12}] \quad (3)$$

and

$$\underline{e}^T = [k_{11} \ k_{22} \ k_{12} \ \epsilon_{11} \ \epsilon_{22} \ \epsilon_{12}] \quad (4)$$

The positive definition of these forces and deformations is illustrated in figure 2.

Normally the matrix \underline{D} cannot be defined directly for complex materials. However, the inverse \underline{D}^{-1} can normally be easily calculated from the basic principles of mechanics or determined experimentally. Therefore, the terms for \underline{D}^{-1} are normally specified as input to a computer program. The numerical values of \underline{D} are then evaluated within the element stiffness subroutine. Hence, the basic behavior of the thin shell, including thermal deformations, is expressed in the following form:

$$\begin{bmatrix} k_{11} \\ k_{22} \\ k_{12} \\ \epsilon_{11} \\ \epsilon_{22} \\ \epsilon_{12} \end{bmatrix} = \begin{bmatrix} P_{11} & P_{12} & P_{13} & C_{11} & C_{12} & C_{13} \\ P_{21} & P_{22} & P_{23} & C_{21} & C_{22} & C_{23} \\ P_{31} & P_{32} & P_{33} & C_{31} & C_{32} & C_{33} \\ C_{11} & C_{21} & C_{31} & D_{11} & D_{12} & D_{13} \\ C_{12} & C_{22} & C_{32} & D_{21} & D_{22} & D_{23} \\ C_{13} & C_{23} & C_{33} & D_{31} & D_{32} & D_{33} \end{bmatrix} \begin{bmatrix} M_{11} \\ M_{22} \\ M_{12} \\ f_{11} \\ f_{22} \\ f_{12} \end{bmatrix} + dT \begin{bmatrix} \alpha_1 \\ \alpha_2 \\ \alpha_3 \\ \alpha_4 \\ \alpha_5 \\ \alpha_6 \end{bmatrix} \quad (5)$$




Figure 1. EXAMPLE OF ANISOTROPIC SHELL




Figure 2. DEFINITION OF POSITIVE FORCES

Or in terms of matrix notation:

$$\underline{\epsilon} = \underline{D}^{-1} \underline{f} + \underline{\epsilon}_0 \quad (5)$$

Where dT is the temperature change and α_1 to α_4 are the measured thermal expansion coefficients. Hence, the thermal forces indicated in equation (2) are calculated from:

$$\underline{f}_0 = - \underline{D} \underline{\epsilon}_0 \quad (6)$$

Each flexibility term in equation (6) has a direct physical meaning. For example, the term C_{12} is the curvature k_{11} due to a unit force, $f_{12} = 1$. Whereas, the term C_{21} is the strain ϵ_{11} at the reference plane caused by the application of a unit bending moment, $m_{11} = 1$. It is apparent that these terms can be best determined experimentally for complex composite materials. Also, the values of these terms are dependent on the definition of the reference plane which must be defined at the same time as the flexibility terms are determined. For the special case of constant thickness isotropic shells the mid-surface is the logical reference plane and the terms C_{1j} and several other flexibility terms are zero in equation (5).

In equation (1) the 6×18 \underline{B} matrix defines the relationship between the deformation terms and the node displacements \underline{v} in the local 1,2,3 system. Or, in matrix form:

$$\underline{\epsilon} = \underline{B} \underline{v} \quad (7)$$

which can be written in submatrix form as

$$\underline{\epsilon}_p = \underline{B}_p \underline{v} \quad (8a)$$

$$\underline{\epsilon}_m = \underline{B}_m \underline{v} \quad (8b)$$

where the "p" and "m" indicate the plate-bending and membrane terms respectively.

BENDING APPROXIMATION - THE DKT ELEMENT

The development of the B_e matrix is based on the standard DKT element [5]. Because the bending and membrane behavior are coupled the DKT formulation will be summarized here in order that the B_e matrix will be developed with consistent approximations in the next section of this report.

The DKT element is based on the independent expansion of the inplane rotations of the reference surface for a 6 node triangle which is shown in figure 3. If the local 1 and 2 directions are indicated by the local x and y coordinates the finite element approximation is written in the following form:

$$\begin{aligned}\beta_x &= a_1 + a_2 x + a_3 y + \frac{1}{2}a_4 x^2 + a_5 xy + \frac{1}{2}a_6 y^2 \\ \beta_y &= a_7 + a_8 x + a_9 y + \frac{1}{2}a_{10} x^2 + a_{11} xy + \frac{1}{2}a_{12} y^2\end{aligned}\tag{9}$$

The six constants a_1 to a_6 , a_7 , can be expressed in terms of the six node rotations β_{x1} to β_{x6} by an inversion of a 6×6 matrix which will produce an equation of the following form:

$$\underline{a}_x = \underline{H} \underline{\beta}_x\tag{10}$$

The same 6×6 matrix, \underline{H} , will relate the constants a_7 to a_{12} , a_y , to the node rotations $\underline{\beta}_y$.

From the theory of thin plates the curvature-displacement relationships are defined by the following equations:


$$k_{xx} = \beta_{x,x} = a_2 + a_4 x + a_6 y\tag{11a}$$

$$k_{yy} = \beta_{y,y} = a_9 + a_{11} x + a_{12} y\tag{11b}$$

$$\begin{aligned}k_{xy} &= \beta_{x,y} + \beta_{y,x} \\ &= a_3 + a_5 x + a_6 y + a_8 + a_{10} x + a_{11} y\end{aligned}\tag{11c}$$


Or, in the following matrix form:

$$\underline{k}_p = \underline{G} \underline{a}\tag{12}$$



(a) BASIC ROTATIONAL UNKNOWNNS

L_{IJ} is the length from I to J



(b) DISPLACEMENTS ALONG TYPICAL SIDE

Figure 3. DISPLACEMENT APPROXIMATION - DKT ELEMENT

Where

$$\underline{a}_p^T = [a_2 + a_3 + a_4 + a_5 + a_6 + a_7 + a_8 + a_{10} + a_{11} + a_{12}] \quad (13)$$

and

$$\underline{G}_p = \begin{bmatrix} 1 & 0 & x & y & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 & 0 & x & y \\ 0 & 1 & 0 & x & y & 1 & 0 & x & y & 0 \end{bmatrix} \quad (14)$$

The six rotational degrees-of-freedom which are associated with a typical side I-J of the element are shown in figure (3a). These rotations can be transformed to a n-s reference system which is parallel and normal to a typical element side as shown in figure (3b). The basic DKT constraints are enforced as follows:

1. The mid-side rotation $\theta_{n,n}$ is set to the average of the values at point I and J. Or,

$$\theta_{n,n} = (\theta_{n,I} + \theta_{n,J}) / 2 \quad (15)$$

2. The s-displacements, above and below the reference plane, and the normal displacement in the z-direction are forced to be cubic functions since the transverse shear strain in the s-direction is set to zero. Therefore, the mid-side normal rotation must satisfy the following equation:

$$\theta_{nm} = - (\theta_{n,I} + \theta_{n,J}) / 4 - 3(w_I + w_J) / (2L_{IJ}) \quad (16)$$

The constraint specified by equation (15) will force the normal displacement along the element sides, above and below the reference plane, to be linear function. Hence, displacement and slope compatibility is satisfied along the sides of all elements. Since no attempt is made to set the transverse shear strains to zero within the element the name Discrete Kirchhoff Triangle was selected as the name of this element.

Equation (15) and (16) can be summarized in matrix form as

$$\begin{bmatrix} \theta_s \\ \theta_N \end{bmatrix} = \begin{bmatrix} 1/2 & 0 \\ 0 & -1/4 \end{bmatrix} \begin{bmatrix} \theta_{s1} \\ \theta_{N1} \end{bmatrix} + \begin{bmatrix} 1/2 & 0 \\ 0 & -1/4 \end{bmatrix} \begin{bmatrix} \theta_{sJ} \\ \theta_{NJ} \end{bmatrix} + 1.5/L_{IJ} \begin{bmatrix} 0 & 0 \\ -1 & 1 \end{bmatrix} \begin{bmatrix} W_I \\ W_J \end{bmatrix} \quad (17)$$

The relationship between the N-S and X-Y coordinate systems are

$$\begin{bmatrix} \theta_s \\ \theta_N \end{bmatrix} = \begin{bmatrix} \cos\delta & \sin\delta \\ -\sin\delta & \cos\delta \end{bmatrix} \begin{bmatrix} \theta_x \\ \theta_y \end{bmatrix} \quad (18a)$$

and

$$\begin{bmatrix} \theta_x \\ \theta_y \end{bmatrix} = \begin{bmatrix} \cos\delta & -\sin\delta \\ \sin\delta & \cos\delta \end{bmatrix} \begin{bmatrix} \theta_s \\ \theta_N \end{bmatrix} \quad (18b)$$

Equation (17) can now be written in the X-Y system as

$$\begin{bmatrix} \theta_x \\ \theta_y \end{bmatrix} = \begin{bmatrix} T1 & T2 \\ T2 & T3 \end{bmatrix} \begin{bmatrix} \theta_{x1} \\ \theta_{y1} \end{bmatrix} + \begin{bmatrix} T1 & T2 \\ T2 & T3 \end{bmatrix} \begin{bmatrix} \theta_{xJ} \\ \theta_{yJ} \end{bmatrix} + 1.5/L_{IJ} \begin{bmatrix} -T_s & T_s \\ T_c & -T_c \end{bmatrix} \begin{bmatrix} W_I \\ W_J \end{bmatrix} \quad (19)$$

Where

$$T1 = 0.5 \cos^2\delta - 0.25 \sin^2\delta$$

$$T2 = 1.5 \sin\delta \cos\delta$$

$$T3 = 0.5 \sin^2\delta - 0.25 \cos^2\delta$$

$$T_s = 1.5 L_{IJ} \sin\delta$$

$$T_c = 1.5 L_{IJ} \cos\delta$$

The six rotational degrees-of-freedom associated with the three mid-side nodes of the triangle can be eliminated by the application of equation (19). These transformations can be summarized by a matrix equation of the following form:

$$\theta = T_p W \quad (20)$$

where T_p is a 12x9 matrix and W represents a 9x1 vector of θ_{x1} , θ_{y1} and w_i for the three nodes of the triangle.

MEMBRANE APPROXIMATION

The membrane behavior of the triangular shell element is based on the basic six node quadratic membrane. If the 12 coefficients are defined by the 12 x 1 vector \underline{b} the membrane strains can be expressed in the following form:

$$\underline{\epsilon}_M = \underline{G}_M \underline{b} \quad (21)$$

As in the case of the DKT plate bending element the three mid-side displacements are rotated to the local N-S coordinate of each side as shown in figure (4). In order to maintain displacement compatibility between element the displacement u_s is assumed to be linear along each side and the displacement u_n is a cubic function. These assumptions can be summarized by the following equations for the displacements at the mid-side nodes:

$$\begin{aligned} u_s &= (u_{s1} + u_{s2})/2 \\ u_n &= (u_{n1} + u_{n2})/2 + L_{12}(\theta_{z1} - \theta_{z2})/8 \end{aligned} \quad (22)$$

These equations can be written in terms of the X-Y coordinate system as

$$\begin{bmatrix} u_x \\ u_y \end{bmatrix} = \begin{bmatrix} .5 & 0 \\ 0 & .5 \end{bmatrix} \begin{bmatrix} u_{x1} \\ u_{y1} \end{bmatrix} + \begin{bmatrix} .5 & 0 \\ 0 & .5 \end{bmatrix} \begin{bmatrix} u_{x2} \\ u_{y2} \end{bmatrix} + .125L_{12} \begin{bmatrix} -\sin\delta & \sin\delta \\ \cos\delta & -\cos\delta \end{bmatrix} \begin{bmatrix} \theta_{z1} \\ \theta_{z2} \end{bmatrix} \quad (23)$$

The six translational displacements at the midside nodes can now be eliminated and three rotational unknowns are introduced at the vertices by the direct application of equation (23). The same basic approach which was used in the DKT formulation is now applied in order to form the matrix equation of the following form:

$$\underline{b} = \underline{T}_M \underline{u} \quad (24)$$

Therefore, the three membrane strains can be written in terms of the nine node displacements as

$$\underline{\epsilon}_M = \underline{G}_M \underline{T}_M \underline{u} \quad (25)$$

It is now possible to evaluate the 24 x 24 element stiffness by the direct application of equation (1).




Figure 5 TRIANGULAR 18 DEGREE-OF-FREEDOM SHELL ELEMENT

COMPUTER PROGRAM IMPLEMENTATION

The 18 x 18 triangular element stiffness matrix, for the element shown in figure 5, is given by equation (1). Where the the strain-displacement matrix \underline{B} can now be written in terms of the bending and membrane submatrices, or

$$\underline{B} = \begin{bmatrix} \underline{G}_r & \underline{0} & \underline{T}_r & \underline{0} \\ \underline{0} & \underline{G}_m & \underline{0} & \underline{T}_m \end{bmatrix} = \underline{G}(x,y) \underline{T} \quad (26)$$

Since the \underline{T} matrix is not a function of x and y it is possible to rewrite equation (1) in the following form:

$$\underline{K} = \underline{T}^T \int \underline{G}^T \underline{D} \underline{G} \, dA \underline{T} \quad (27)$$

The matrix \underline{G} is very sparse and contains only the terms 1, x and y; therefore the integral can be evaluated directly with a minimum of numerical effort as illustrated by the FORTRAN listing given in Appedix A. In addition, an integral reduction factor can be used on selective terms in order to improve the membrane performance as suggested in reference (3).

NUMERICAL EXAMPLES

In order to illustrate the behavior of the element and to compare the results with other shell elements several examples will be presented.

Cantilever Beam

The beam shown in figure 6 is idealized by a 1x4 rectangular mesh and is subjected to a load of 40 kips at the tip of the cantilever. The theoretical displacement at the tip, including shearing deformation, is 0.3558 inches.

The Taylor quadrilateral element shell yields a displacement of 0.3467 inches; or an error of -1.02 percent. Note that the rotations at the base of the cantilever are set to zero which is inconsistent with the existence of shearing deformations.

The element presented in this report, TSHELL, was used to model this beam with two triangles to form each quadrilateral. The completely integrated element produces a displacement of 0.2695 inches, or an error of -24.3 percent. With a reduced integration factor of 0.5 the tip displacement is 0.3726, or an error of +4.5 percent.

For all example problems presented in this report a reduced integration factor of 0.5 is used. The use of the reduced integration factor has the major advantage over one point integration is that a "rank deficiency" is not introduced into the element.





Figure 6. CANTILEVER BEAM EXAMPLE

Twisted Beam

The twisted beam shown in figure 7 has become a standard test problem for thin shell elements [6]. Table 1 summarizes the results obtained using four different elements. It is important to note that the SHELL element does not converge as the mesh is refined. The reason for this unacceptable behavior is because the element is flat and it cannot model the twisted surface accurately. The new triangular element gives good results for this problem.



WIDTH = 1.1
DEPTH = 0.32
TWIST = 90°
 $E = 29.0 \times 10^6$
POISSON RATIO = 0.22
MESH = 12 x 2
Unit loads at end


Figure 7. TWISTED BEAM EXAMPLE

Table 1. Deflection Under Load For Twisted Beam

Element Type and Mesh	IN-PLANE	error	OUT-OF-PLANE	error
Reference Values	0.005424	0	0.001754	0
NASTRAN QUAD4	0.005386	-0.7	0.001727	-1.5
NASTRAN QUAD8	0.005413	-0.2	0.001750	-0.2
SHELL (2x12)	0.005779	+6.4	0.001993	+13.7
" (4x24)	0.006849	+26.2	0.002750	+56.9
TSHELL (2x12)	0.005390	-0.6	0.001717	-2.1
" (4x24)	0.005399	-0.5	0.001735	-1.1

Scordelis-Lo Roof

The shell structure shown in figure 8 is a standard test problem [6]. Table 2 summarizes the maximum displacement obtained using different elements and meshes. For this problem the SHELL element results are very good. However, the TSHELL results appear to converge very slowly. However, for even the coarse mesh, the results are of acceptable accuracy for normal engineering analysis.



THICKNESS = 0.25
 $E = 4.32 \times 10^8$
 POISSON RATIO = 0.0
 LOADING: 90/unit area in z dir.
 $u = w = 0$ on curved edge
 MESH $N \times N$ on quadrant

Figure 8. SCORDELIS-LO CYLINDRICAL SHELL

Table 2. Maximum Displacement Of Scordelis-Lo Roof

Element Type and Mesh		VALUE	ERROR
Reference Value		0.3086 ft.	0.0 %
SHELL	N=4		+5.2 %
	N=6		+1.7 %
	N=8		+0.6 %
	N=10		+0.2 %
TSHELL	N=4	0.3036	-1.6 %
	N=8	0.2982	-3.4 %
	N=12	0.2997	-2.9 %

Spherical Shell

A spherical shell subjected to point loads is shown in figure 9 and is also a standard test problem [6]. This structure clearly illustrates the weakness of the TSHELL element. With a 12x12 mesh the error in displacement is 17.8 percent. The reason for this very slow convergence is that the triangular elements essentially add rib reinforcement to the very flexible spherical surface.




Figure 9. SPHERICAL SHELL EXAMPLE

Table 3. Displacement of Spherical Shell

Element Type and Mesh	VALUE	ERROR
Reference Value	0.0925 in.	0.0 %
SHELL N=4		-46.1 %
N=8		-3.3 %
N=12		-0.7 %
TSHELL N=8	0.051265	-44.6 %
N=12	0.075974	-17.8 %

FINAL REMARKS

A new anisotropic triangular shell element, with normal rotations at the nodes has been developed. The element can be connected directly to nodes with beam elements without special consideration.

The general accuracy of the element has been demonstrated. Care should be taken if the element is used to model shells which have double curvature.

REFERENCES

1. Carpenter, N., H. Stolarski and T. Belytschko, "A Flat Triangular Shell Element With Improved Membrane Interpolation", Communications in Applied Numerical Methods, Vol. 1, pp. 161-168, (1985).
2. Taylor, R.L. and J.C. Simo, "Bending and Membrane Elements for The Analysis of Thick and Thin Shells", Proceedings of The NUMETA-85 Conference, Swansea, January 1985.
3. Bergan, P.G. and C.A. Felippa, "A Triangular Membrane Element with Rotational Degrees-of-Freedom", Computer Methods in Applied Mechanics and Engineering, Vol. 50, pp. 25-69, (1985).
4. Stander, N., "Finite Element Analysis of Prismatic Structures", Ph.D. Dissertation, University of California, Berkeley, August 1986.
5. Batoz, J.L., "An Explicit Formulation for an Efficient Triangular Plate Bending Element", IJNME, Vol. 18, pp. 1077-1089, (1982).
6. MacNeal, R.H. and R.L. Harder, "A Proposed Standard Set of Problems to Test Finite Element Accuracy", Finite Elements in Analysis and Design, Vol. 1, pp. 3-20, (1985).

APPENDIX A - FORTRAN LISTING OF TRIANGULAR SHELL ELEMENT

```

C-----SHELLT
SUBROUTINE SHELLT(S,T,F,D,C,V,X,Y,AREA,PRES,TEMP)
IMPLICIT REAL*8 (A-H,O-Z)
C----- INFORMATION CALCULATED -----
C      S = 18 x 18 STIFFNESS MATRIX IN X-Y-Z SYSTEM
C      T = 18 x 18 FORCE-TRANSFORMATION MATRIX
C      F = 18 x 2 FORCES DUE TO PRESSURE AND TEMP.
C----- INFORMATION SPECIFIED -----
C      D = 6 x 6 MATERIAL PROPERTY MATRIX
C      LOCAL FORCES ARE ASSUMED TO BE IN THE
C      ORDER M11, M22, M12, N11, N22 and N12
C      C = 6 x 1 MATRIX OF THERMAL EXPANSION TERMS
C      V = THE DIRECTION COSINE ARRAY WHICH RELATES THE
C      LOCAL 1-2-3 TO THE GLOBAL X-Y-Z SYSTEM
C      X, Y, Z GLOBAL NODE COORDINATES
C      AREA = ELEMENT AREA
C      PRES = AVERAGE SURFACE PRESSURE
C      TEMP = AVERAGE TEMPERATURE CHANGE
C-----
C      WRITTEN BY EDWARD L. WILSON, JAN. - JUNE 1987
C-----
      DIMENSION V(4,4),KL(24,3),XY(3),E(6),H(6,6),
      . S(18,18),T(18,18),D(6,6),X(6),Y(6),AP(10,12),
      . AM(10,12),A(20,18),AIN(3,3),G(20,20),JP(9),
      . JM(9),TT(20,18),F(18,2),C(6)
      DATA
      . JP /4,5,10,11,16,17,3,9,15/,
      . JM /1,2,7,8,13,14,6,12,18/
      DATA KL /
      . 1,1,1, 2,2,2, 3,3,3,3,3,3, 4,4,4, 5,5,5, 6,6,6,6,6,6,
      . 1,3,4, 7,9,10, 2,4,5,6,8,9, 11,13,14, 17,19,20,12,14,15,16,18,19,
      . 1,2,3, 1,2,3, 1,2,3,1,2,3, 1,2,3, 1,2,3, 1,2,3,1,2,3/
      NT = 24
C----- CHANGE TO LOCAL COORDINATES SYSTEM -----
      XC = ( X(1) + X(2) + X(3) ) / 3.
      YC = ( Y(1) + Y(2) + Y(3) ) / 3.
      DO 20 I=1,3
      X(I) = X(I) - XC
      Y(I) = Y(I) - YC
      DO 20 J=1,3
20  AIN(I,J) = 0.0
C----- COMPUTE INTEGRALS -----
      RED = 0.50*AREA
      AIN(1,1) = AREA
      AIN(2,2) = -RED*( X(1)*X(2) + X(2)*X(3) + X(3)*X(1) ) / 6.
      AIN(2,3) = RED*( X(1)*Y(1) + X(2)*Y(2) + X(3)*Y(3) ) / 12.
      AIN(3,3) = -RED*( Y(1)*Y(2) + Y(2)*Y(3) + Y(3)*Y(1) ) / 6.
      AIN(3,2) = AIN(2,3)

```

FORTTRAN LISTING OF TRIANGULAR SHELL ELEMENT

```

C----- FORM "H" ARRAY FOR GENERAL 6-NODE TRIANGLE -----
      CALL FORMH(H,A,X,Y)
C----- FORM "AP" ARRAY FOR PLATE ELEMENT -----
      CALL FORMAP(AP,H,X,Y)
C----- FORM "AM" ARRAY FOR PLANE ELEMENT -----
      CALL FORMAM(AM,H,X,Y)
C----- FORM 20x20 "G" ARRAY -----
      DO 160 K=1,20
      DO 160 L=1,K
160    G(K,L) = 0.0
C
      DO 200 N=1,NT
      I  = KL(N,1)
      K  = KL(N,2)
      II = KL(N,3)
      DO 190 M=1,NT
      J  = KL(M,1)
      IF (D(I,J).EQ.0.0) GO TO 190
      L  = KL(M,2)
      IF (L.GT.K) GO TO 190
      JJ = KL(M,3)
      G(K,L) = G(K,L) + D(I,J)*AIN(II,JJ)
190    CONTINUE
200    CONTINUE
C
      DO 210 K=1,20
      DO 210 L=1,K
210    G(L,K) = G(K,L)
C----- FORM 20x18 "A" ARRAY -----
      DO 300 J=1,18
      F(J,1) = 0.0
      F(J,2) = 0.0
      DO 300 I=1,20
300    A(I,J) = 0.0
C
      DO 305 J=1,9
      JJ = JP(J)
      DO 305 I=1,10
305    A(I,JJ) = AP(I,J)
C
      DO 310 J=1,9
      JJ = JM(J)
      DO 310 I=1,10
310    A(I+10,JJ) = AM(I,J)
C----- ROTATE TO GLOBAL X, Y, Z SYSTEM -----
      DO 350 N=1,6
      NZ = 3*N
      NY = NZ - 1
      NX = NY - 1
      DO 350 I=1,20
      XX = A(I,NX)*V(1,1) + A(I,NY)*V(1,2) + A(I,NZ)*V(1,3)
      YY = A(I,NX)*V(2,1) + A(I,NY)*V(2,2) + A(I,NZ)*V(2,3)
      ZZ = A(I,NX)*V(3,1) + A(I,NY)*V(3,2) + A(I,NZ)*V(3,3)
      A(I,NX) = XX
      A(I,NY) = YY
350    A(I,NZ) = ZZ

```

FORTTRAN LISTING OF TRIANGULAR SHELL ELEMENT

```

C----- FORM 18x18 ELEMENT STIFFNESS -----
      DO 380 I=1,20
      DO 380 J=1,18
      CALL DOTP(G(1,I),A(1,J),SUM,20)
      380 TT(I,J) = SUM
C
      DO 400 I=1,18
      DO 400 J=1,I
      CALL DOTP(A(1,I),TT(1,J),SUM,20)
      S(I,J) = SUM
      400 S(J,I) = SUM
C----- FORM FORCE-DISPLACEMENT TRANSFORMATION ARRAY -----
      NO = 0
      DO 500 N=1,3
      XY(1) = 1.0
      XY(2) = X(N)
      XY(3) = Y(N)
C
      DO 450 K=1,18
C
      DO 410 I=1,6
      410 E(I) = 0.0
C
      DO 420 M=1,NT
      I = KL(M,1)
      L = KL(M,2)
      J = KL(M,3)
      420 E(I) = E(I) + XY(J)*A(L,K)
C
      DO 440 I=1,6
      SUM = 0.0
      DO 430 J=1,6
      430 SUM = SUM + D(J,I)*E(J)
      II = NO + I
      440 T(II,K) = SUM
C
      450 CONTINUE
C
      500 NO = NO + 6
C----- FORM THERMAL FORCES -----
      IF (TEMP.EQ.0.0) GO TO 625
      DO 600 I=1,6
      CALL DOTP(D(1,I),C(1),E(I),6)
      600 CONTINUE
      DO 610 I=1,6
      610 C(I) = - TEMP*E(I)
C
      DO 620 I=1,18
      F(I,2) = C(1)*A(1,I)
      F(I,2) = C(2)*A(7,I)
      F(I,2) = C(3)*( A(2,I) + A(6,I) )
      F(I,2) = C(4)*A(11,I)
      F(I,2) = C(5)*A(17,I)
      620 F(I,2) = C(6)*( A(12,I) + A(16,I) )

```


FORTRAN LISTING OF TRIANGULAR SHELL ELEMENT

```

C----- CALCULATE PRESSURE FORCES -----
625 IF(PRES.EQ.0.0) GO TO 800
    FORCE = - AREA*PRES/3.0
    DO 630 I=1,18,6
        F(I,1) = FORCE*V(1,3)
        F(I+1,1) = FORCE*V(2,3)
630 F(I+2,1) = FORCE*V(3,3)
C
800 RETURN
C-----
    END
C----- FORMH
    SUBROUTINE FORMH(H,B,X,Y)
    IMPLICIT REAL*8 (A-H,O-Z)
    DIMENSION H(6,6),B(6,6),Y(4),X(4)
C----- FORM COEFFICIENT MATRIX FOR 6 NODE TRIANGLE -----
    X(4) = ( X(1) + X(2) ) / 2.
    X(5) = ( X(2) + X(3) ) / 2.
    X(6) = ( X(3) + X(1) ) / 2.
    Y(4) = ( Y(1) + Y(2) ) / 2.
    Y(5) = ( Y(2) + Y(3) ) / 2.
    Y(6) = ( Y(3) + Y(1) ) / 2.
C
    DO 100 I=1,6
        H(1,I) = 1.0
        H(2,I) = X(I)
        H(3,I) = Y(I)
        H(4,I) = X(I)*X(I) / 2.
        H(5,I) = X(I)*Y(I)
100 H(6,I) = Y(I)*Y(I) / 2.
C----- INVERT TO FORM H MATRIX -----
    DO 200 I=1,6
        DO 200 J=1,I
            SUM = 0.0
            DO 190 K=1,6
190 SUM = SUM + H(I,K)*H(J,K)
            B(J,I) = SUM
200 B(I,J) = SUM
C
    CALL SYMSOL(B,H,6,6,0)
C
    RETURN
    END

```

FORTRAN LISTING OF TRIANGULAR SHELL ELEMENT

```

C----- FORMAP
      SUBROUTINE FORMAP(AP,H,X,Y)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION AP(10,12),H(6,6),Y(4),X(4),IT(4)
      DATA IT /1,3,5,1/
C----- FORM 12 DOF MATRIX -----
      DO 240 I=2,6
      K = 0
      DO 240 J=1,6
      K = K + 1
      AP(I-1,K) = 0.0
      AP(I+4,K) = H(I,J)
      K = K + 1
      AP(I-1,K) = - H(I,J)
240   AP(I+4,K) = 0.0
C----- ELIMINATION OF 4,5,6 MID-SIDE ROTATIONS -----
      X(4) = X(1)
      Y(4) = Y(1)
      DO 300 N=1,3
      DX = X(N+1) - X(N)
      DY = Y(N+1) - Y(N)
      XL = DSQRT( DX*DX + DY*DY )
      S = DY / XL
      C = DX / XL
      T1 = C*C/2. - S*S/4.
      T2 = 0.75*S*C
      T3 = S*S/2. - C*C/4.
      XX = 1.5/XL
      TC = XX*C
      TS = XX*S
C
      I1 = IT(N)
      I2 = I1 + 1
      J1 = IT(N+1)
      J2 = J1 + 1
      L2 = 6 + N + N
      L1 = L2 - 1
      I = N + 6
      J = I + 1
      IF (N.EQ.3) J = 7
C
      DO 300 K=1,10
      AP(K,I1) = AP(K,I1) + AP(K,L1)*T1 + AP(K,L2)*T2
      AP(K,I2) = AP(K,I2) + AP(K,L1)*T2 + AP(K,L2)*T3
      AP(K,J1) = AP(K,J1) + AP(K,L1)*T1 + AP(K,L2)*T2
      AP(K,J2) = AP(K,J2) + AP(K,L1)*T2 + AP(K,L2)*T3
      W = - AP(K,L1)*TS + AP(K,L2)*TC
      AP(K,L1) = 0.0
      AP(K,L2) = 0.0
      AP(K,I) = AP(K,I) + W
      AP(K,J) = AP(K,J) - W
300   CONTINUE
C
      RETURN
      END

```

FORTRAN LISTING OF TRIANGULAR SHELL ELEMENT

```

C----- FORMAM
      SUBROUTINE FORMAM(AM,H,X,Y)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION AM(10,12),H(6,6),Y(4),X(4),IT(4)
      DATA IT /1,3,5,1/
C----- FORM 12 DOF MATRIX -----
      DO 240 I=2,6
      K = 0
      DO 240 J=1,6
      K = K + 1
      AM(I-1,K) = H(I,J)
      AM(I+4,K) = 0.0
      K = K + 1
      AM(I-1,K) = 0.0
      240 AM(I+4,K) = H(I,J)
C----- ROTATE NODE 4, 5, 6 MID-SIDE DISPLACEMENTS ----
      X(4) = X(1)
      Y(4) = Y(1)
C
      DO 300 N=1,3
      DX = X(N+1) - X(N)
      DY = Y(N+1) - Y(N)
C
      NY = 2*N + 6
      NX = NY - 1
      IX = IT(N)
      IY = IX + 1
      JX = IT(N+1)
      JY = JX + 1
      I = N + 6
      J = I + 1
      IF (N.EQ.3) J = 7
C----- ELIMINATE MID-SIDE DISPLACEMENTS -----
      DO 260 K=1,10
      TT = .125*( - AM(K,NX)*DY + AM(K,NY)*DX )
      AM(K,IX) = AM(K,IX) + AM(K,NX)/2.
      AM(K,IY) = AM(K,IY) + AM(K,NY)/2.
      AM(K,JX) = AM(K,JX) + AM(K,NX)/2.
      AM(K,JY) = AM(K,JY) + AM(K,NY)/2.
      AM(K,NX) = 0.0
      AM(K,NY) = 0.0
      AM(K,I) = AM(K,I) + TT
      AM(K,J) = AM(K,J) - TT
      260 CONTINUE
C
      300 CONTINUE
C
      RETURN
      END

```

FORTTRAN LISTING OF TRIANGULAR SHELL ELEMENT

```

C----- LOCALT
      SUBROUTINE LOCALT(XYZ,X,Y,V,AREA,IAXIS)
      IMPLICIT REAL*8 (A-H,O-Z)
      DIMENSION XYZ(3,3),X(6),Y(6),V(4,4)
C---- TRANSFORM TO LOCAL COORDINATE SYSTEM -----
C      WRITE (*,3000) XYZ
C 3000 FORMAT (3F15.4)
      IF (IAXIS.NE.0) THEN
        DO 100 I=1,3
          V(I,4) = 0.0
          I = IABS(IAXIS)
          V(I,4) = IAXIS/I
        END IF
C---- DEFINE LOCAL 1,2,3 REFERENCE SYSTEM-----
      CALL VECTOR(V(1,1),XYZ(1,1),XYZ(1,2),XYZ(1,3),
        XYZ(2,1),XYZ(2,2),XYZ(2,3))
      CALL VECTOR(V(1,2),XYZ(1,1),XYZ(1,2),XYZ(1,3),
        XYZ(3,1),XYZ(3,2),XYZ(3,3))
      CALL CROSS(V(1,1),V(1,2),V(1,3))
      IF (IAXIS.NE.0) CALL CROSS(V(1,4),V(1,3),V(1,1))
      CALL CROSS(V(1,3),V(1,1),V(1,2))
C
      DO 5 N=1,3
        X(N) = XYZ(N,1)*V(1,1) + XYZ(N,2)*V(2,1) + XYZ(N,3)*V(3,1)
        5 Y(N) = XYZ(N,1)*V(1,2) + XYZ(N,2)*V(2,2) + XYZ(N,3)*V(3,2)
C---- CALCULATE AREA OF ELEMENT -----
      AREA = ( X(2)*Y(3) - X(3)*Y(2) + X(3)*Y(1)
        - X(1)*Y(3) + X(1)*Y(2) - X(2)*Y(1) ) / 2.0
      IF (AREA.LE.0.0) THEN
        PAUSE ' ZERO OR NEGATIVE AREA '
        RETURN
      END IF
C-----
      RETURN
      END

```

INITIAL DISTRIBUTION LIST

	No. Copies
1. Defense Technical Information Center Cameron Station Alexandria, VA 22304-6145	2
2. Research Administration, (Code 012) Naval Postgraduate School Monterey, CA 93943-5000	1
3. Library, (Code 0142) Naval Postgraduate School Monterey, CA 93943-5000	2
4. Professor Anthony J. Healey, Code 69Hy Chairman, Mechanical Engineering Department Naval Postgraduate School Monterey, CA 93943-5000	1
5. Professor Gilles Cantin, Code 69Ci Mechanical Engineering Department Naval Postgraduate School Monterey, CA 93943-5000	2
6. Mr. Joseph Corado, Code 1702 David W. Taylor Naval Ship Research and Development Center Bethesda, MD 20034	2
7. Professor Edward Wilson Mechanical Engineering Department Naval Postgraduate School Monterey, CA 93943-5000	1